
*                 |     _|_          |
*                 |    /_  \ B2      | H_Field
*                 |___// \  \        |
*                 |   \   \//        |    ^ B_Field
*                 |    \___/H_Field2 |   /_\
*      Vtime      |      |____       |    |____||__/\  ____
*        ____     |     _|_  _|_     |    |    ||    \/    |
*      _|_   | B1_|_  D1 ^   \ /D2   |   _|_     V_coil   _|_
*  VT /_  \  |  /_  \   /_\  _V_     |__/_  \B3           ///
*    // \  \ |_// \  \   |____|________// \  \   V= dB/dt
*    \   \//   \   \//   |  MagMemH    \   \//
*     \___/     \___/   _|_             \___/
*       |         |     ___               |
*      _|_       _|_     | C1            _|_
*      ///       ///    _|_              ///
*                       ///
VT     Vtime    0        PWL ( 0 0 1 1 )   
*B1    H_Field  0        V =   2* v(Vtime )*asin(sin(3.142*10*v(Vtime)))
B1     H_Field  0        V =   2* asin(sin(3.142*50*v(Vtime)))
B2     H_Field  H_Field2 V =   .8*v(H_Field)
D1     H_Field2 MagMemH  DD
D2     MagMemH  H_Field2 DD
C1     MagMemH  0        1p
B3     B_Field  0        V =   tanh(V(H_Field)+2*V(MagMemH))
C2     B_Field  V_coil   1u
R2     V_coil   0        1
.tran  .1m      .1       0     .1m   
.control
run
set    pensize = 1
plot   v(h_field) v(magmemh) v(b_field) v(v_coil)*1000
plot   v(b_field) vs  v(h_field)
.endc
.model  DD D(IS=3.15e-18)
.end

The need is to somehow store what happens in
the past which in this case the magnetic memory.



The store magnetic memory is used to offset
the applied H_field.

The offseted H_Field can then be put into
some nonlinear function. (tanh in this case)

Using diodes and a capacitor with a variable



offset may still needs some tweeking.

But it is not obvious why enough circuitry couldn't be
added to achieve the needed correlation to a real BH curve. 


